# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>

pkgname=gnome-themes-standard
pkgver=3.22.3
pkgrel=1
pkgdesc="Standard Themes for GNOME Applications"
url="https://git.gnome.org/browse/gnome-themes-standard"
arch=(i686 x86_64)
license=(GPL2)
depends=(cantarell-fonts ttf-dejavu librsvg)
makedepends=(intltool gtk2 gtk3 gnome-common git)
optdepends=('gtk-engines: HighContrast GTK2 theme')
groups=(gnome)
options=(!emptydirs)
_commit=015785c53c37912bc6266c9cf107a0d1c8816d51  # tags/3.22.3^0
source=("git+https://git.gnome.org/browse/gnome-themes-standard#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/-/+/g'
}

prepare() {
  cd $pkgname
  NOCONFIGURE=1 ./autogen.sh
}

build() {
  cd $pkgname
  ./configure --prefix=/usr
  sed -i -e 's/ -shared / -Wl,-O1,--as-needed\0/g' libtool
  make
}

package() {
  cd $pkgname
  make DESTDIR="${pkgdir}" install
}

# vim:set ts=2 sw=2 et:
